
Инструкция по работе сенсорных панелей DOP с GSM-модемами 

 
 
 

Введение 
 
 
Совместное использование сенсорной панели и GSM-модема предоставляет для пользователя 
простой и удобный способ реализации дистанционного управления системой, частью 
которой является панель оператора, а также возможность получать от панели SMS-
сообщения при возникновении соответствующих ситуаций на объекте, где она установлена. 
В качестве передающей среды используется сотовая связь, что дает возможность применять 
обычный сотовый телефон в качестве терминала для общения с панелью.  
 
 
 
 
 

Описание аппаратного интерфейса сенсорных панелей DOP и GSM-модемов 
 
 
В зависимости от модели, сенсорные панели семейства DOP имеют 2 или 3 
коммуникационных порта. Порт СОМ1 всегда является RS232 с поддержкой аппаратного 
управления потоком (RTS/CTS), а порты СОМ2 и СОМ3 могут переключаться с интерфейса 
RS232 на RS485.  
Порты всегда располагаются на тыльной стороне панели: 

В качестве соединителя используется стандартный разъем DB9, и только у панелей DOP-AS 
используется на СОМ2 и СОМ3 винтовой клеммник.  
 
Порт СОМ1 панелей семейства DOP-A имеет разъем DB-9F (мама), а у панелей DOP-B 
разъем DB-9M (папа).  
 
Все порты панелей могут работать в активном режиме, т.е. получать и отправлять сообщения. 
Следовательно, GSM-модем можно подключить к любому из них.  
 
Для работы с панелью оператора подойдет любой GSM-модем, имеющий порт RS232 или 
RS485, а также поддерживающий стандартный набор АТ-команд. Но для российского рынка 
наиболее оптимальными со всех точек зрения являются GSM-модемы отечественной 
разработки и сборки семейства Teleofis. Панели DOP тестировались именно с ними.  
 
В зависимости от того, к какому порту панели будет подключен GSM-модем, можно выбрать 
модель RX100-R COM с портом RS232 или RX108-R RS485 с портом RS485 (антенна и 
источник питания в комплект не входят).  

Если предполагается подключать GSM-модем к панели оператора DOP-A+ по COM1 RS232, 
то необходимо спаять переходник, так как и модем и панель имеют разъемы DB-9F (мама).  
 
 

 
 
 
У переходника разъемы DB-9M (папа) с обеих сторон с прямой распайкой: 
 

 
 
 
При работе через СОМ2 и СОМ3 в 3-х проводном режиме RS232 в переходнике необходимо 
перемкнуть контакты 7 и 8 в обоих разъемах: 
 
 

 
 
 
При работе по RS485 распайка в соответствии с Руководством на панели оператора.  

Таким образом, общая схема соединения панели оператора и GSM-модема выглядит 
следующим образом: 
 

 
 
 
 
 

Базовая настройка оборудования 
 
 
Для организации связи все оборудование должно быть настроено определенным образом. 
Также, SIM-карта должна иметь тариф с поддержкой SMS-сообщений.  
 
Функция проверки PIN-кода на SIM-карте должна быть отключена!!! Это можно сделать, 
вставив SIM-карту в любой сотовый телефон. 
 
 
Настройка СОМ порта панели оператора DOP 
 
 
Коммуникационные настройки панели осуществляются в меню Options – Configuration  

Далее во вкладке Communication у по рта, к кото р ому будет подключен GSM-модем, 
необходимо выбрать нулевой драйвер «NULL», так как панель будет фо рмир о вать запросы 
путем макрокоманд.  
 

 
 
Нажмите «ОК». 
 
Настройку панели на этом можно считать законченной.  
 
 

1. ATS0=1&W (данная команда устанавливает ответ после первого звонка) 

Настройка рабочих параметров GSM-модема, подключенного к панели 
 
Данная процедура называется инициализацией модема и заключается в передаче GSM-
модему следующих АТ-команд: 
 

2. ATE0&W (выключить эхо) 
3. AT+IPR=9600&W (устанавливает скорость обмена 9600 б/с, формат данных по 

умолчанию фиксированный 8, N, 1)  

4. AT+CMGF=1&W (включает текстовый режим) 
5. AT+CSCS=”IRA” (установка кодировки ASCII) 
6. AT+CSMP=17,167,0,241 (установка параметров текстового режима для ASCII) 
7. AT+CSCA? (данная команда выводит номер сервисного центра SMS-сообщений. Для 

МТС ответ должен быть + 7 916 899 91 00. Номер можно посмотреть и вставив SIM-
карту в сотовый телефон: Меню – Сообщения – Параметры – Сообщение SMS и пункт 
Центр Обслуживания. Как правило, номер сервисного центра всегда по умолчанию 
находится на SIM-карте. Если вдруг окажется, что его на карте нет, то его можно 
узнать у поставщика услуг сотовой связи и ввести следующим образом, например для 
МТС: AT+CSCA=”+79168999100”,145) 

8. AT+CNMI=0,0,0,0,0 (отключает несанкционированную выдачу модемом сообщения о 
получении SMS) 

9. AT&W (запись введенных выше параметров в энергонезависимую память. Данную 
команду нужно обязательно ввести. В противном случае при пропадании питания все 
параметры сбросятся на заводские) 

 
Инициализацию модема можно осуществить из программной среды для контроллеров 
WPLSoft (или любой подходящей терминальной программки): 
 
Вкладка Options –> Modem Connection: 
 

 
 
Установите флажок «Init Modem with AT Command». Команды вводятся заглавными буквами 
без пробелов в строке «AT Instruction». Символ «CR» в конце команды ставить не надо 
(программа подставляет его автоматически). Для отправки АТ-команды в GSM-модем 
нажмите кнопку «Send». Отправляемая команда появится в поле «Sending Data», а ответ 
модема в поле «Receiving Data». 

Порядок написания коммуникационных макросов для панелей оператора DOP 
 
 
Для создания коммуникационного макроса будут полезны следующие команды: 
 
 
INITCOM – инициализация СОМ порта 
SELECTCOM – выбор СОМ порта 
CLEARCOMBUFFER – очистка буфера данных СОМ порта 
PUTCHARS – отправка последовательности символов в буфер отправки СОМ порта 
GETCHARS – получение содержимого из буфера приема СОМ порта 
CHR – конвертация символов в коды ASCII 
SWAP – перестановка местами байтов в регистре 
MOV – перемещение данных из одного регистра в другой 
BMOV – перемещение блока данных из последовательности регистров в другую 
последовательность регистров 
W2B – извлечение младшего байта (8 бит) из регистра данных (слово 16 бит) 
 
Подробное описание данных команд можно найти в Руководстве по программированию 
панелей оператора семейства DOP. Коммуникационные блоки лучше писать в составе 
фонового макроса.  
 
 
 
Выбор и инициализация СОМ порта панели 
 
Для отправки команды в GSM-модем сначала необходимо определить путь и формат данных.  
 
Команда SELECTCOM определяет через какой порт будет осуществляться связь: 
 

 
 
0 – СОМ1 
1 – СОМ2 
2 – СОМ3 

Команда INITCOM устанавливает формат данных (протокол). Необходимо установить 
протокол, чтобы он соответствовал тому, что был записан в GSM-модем при его 
инициализации:  
 

 
 
Получится команда:   $100 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
(если был выбран порт СОМ1) 
 
Команды SELECTCOM и INITCOM должны записываться перед каждой отправляемой в 
СОМ порт последовательностью символов.  
 
 
 

Шаг 1. При помощи команды CHR преобразовать АТ-команду с номером телефона и текст 
сообщения в коды ASCII. Здесь необходимо учитывать, что после преобразования соседние 

Отправка SMS-сообщений панелью оператора  
 
Сообщение может быть отправлено в кодировке ASCII (режим IRA),  то гда получатель на 
экране сотового телефона может видеть только латинские буквы и цифры. Сообщение может 
быть отправлено в UNICODE (режим UCS2), тогда появляется возможность видеть на экране 
телефона русские буквы (и любые другие), но размер строки, содержащей сообщение, 
увеличивается в 4 раза по сравнению с тем же сообщением, которое сделано в кодировке 
ASCII. Это связано с тем, что символ в UNICODE кодируется четырьмя символами ASCII. 
Также, необходимо учитывать, что в UNICODE можно будет отправить максимум 16 русских 
символов. 
 
Если предполагается отправлять русские символы, то в макросе необходимо предусмотреть 
отправку соответствующих АТ-команд в GSM-модем, которые переведут его в режим работы 
в кодировке UNICODE.  
 
Если предполагается работать только с латинскими символами, то для настройки GSM-
модема достаточно процедуры инициализации, описанной выше.  
 
Макрос на отправку SMS в кодах ASCII (см. Приложение 1) необходимо реализовывать по 
следующей схеме: 

символы меняются местами (старший и младший байты). Количество символов должно быть 
четным. Т.е. при необходимости в конце добавляется пробел, который также является 
символом. Символы «СR» и «SUB» нео бходимо  записывать отдельно ,  так как о ни не 
конвертируются.  
 
Например: 
CHR($20, "AT+CMGS="89163769831" ") 
$31 = 000DH 
 
Символ СR (000DH) записан наоборот, так как команда PUTCHARS при пересылке данных 
меняет местами старший и младший байты.  
 
Шаг 2. При помощи команды PUTCHARS послать в GSM-модем АТ-команду на отправку 
SMS-сообщения и номера телефона абонента. Длина отправляемого сообщения в команде 
PUTCHARS задается в байтах. 
 
Шаг 3. После задержки таймера в 1 сек. (Delay 1000) при помощи команды PUTCHARS 
переслать в GSM-модем текст сообщения в кодах ASCII, конец сообщения обозначается 
символом SUB (SUB в HEX = H1A) 
 
Ниже приводится пример отправки сообщения «WATER-PUMP ON» на телефон 89163769831 
через СОМ2: 
 
$130 = INITCOM(1, 2, 1, 0, 0, 6, 0) 
SELECTCOM(1) 
CHR($20, "AT+CMGS="89163769831" ") 
$31 = 000DH 
CHR($50, "WATER-PUMP ON ") 
$57 = 001AH 
$70 = PUTCHARS($20, 24, 500) 
Delay(1000) 
$90 = PUTCHARS($50, 16, 500) 
 
 
Если предполагается отправлять SMS-сообщения русскими символами, необходимо сначала 
перевести GSM-модем в режим UCS2 и установить его параметры, потом записать АТ-
команду прямым образом, а телефонный номер и текст сообщения в UNICODE (см. 
Приложение 2), а затем преобразовать в коды ASCII командой CHR. Далее используя команду 
PUTCHARS отправить в GSM-модем.  
 
Ниже приводится пример отправки SMS-сообщения «Обогрев ВКЛ» на тел. 89163769831 
через СОМ1:  
 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($170, "AT+CSCS="UCS2"") 
$177 = 000DH 
CHR($190, "AT+CSMP=17,167,0,25 ") 
$200 = 000DH 
$280 = PUTCHARS($170, 16, 500) 
Delay(1000) 
$300 = PUTCHARS($190, 22, 500) 

CHR($210, "AT+CMGS="00380039003100360033003700360039003800330031"") 
$237 = 000DH 
CHR($250, "041E0431043E043304400435043200200412041A041B") 
$272 = 001AH 
$320 = PUTCHARS($210, 56, 500) 
Delay(1000) 
$340 = PUTCHARS($250, 46, 500) 
 
 
Обработка панелью оператора входящих SMS-сообщений 
 
 
Входящие SMS-сообщения сначала принимаются GSM-модемом и сохраняются на SIM-
карте, последовательно накапливаясь в ее регистрах. Однако панель оператора не может 
считывать все регистры SIM-карты, так как это программно практически не реализуемо, а 
главное дальнейшая обработка становится чрезвычайно трудной.  
Исходя из этого, нужно читать только первый регистр на SIM-карте, а потом 
соответствующей АТ-командой сразу стирать его содержимое. Таким образом, остается 
необходимость читать только один регистр, что очень удобно для дальнейшей обработки 
полезной информационной составляющей полученного SMS-сообщения. Так как первый 
регистр будет всегда свободен, то входящие сообщения будут поступать только туда.  
 
Для сокращения объема программы и облегчения дальнейшей обработки GSM-модем лучше 
перевести в режим ASCII, а SMS-сообщение должно содержать только цифры.  
 
Для обработки входящих SMS-сообщений используются следующие АТ-команды: 
 
AT+CMGR=1CR  –  чтение входящего сообщения в 1-м регистре SIM-карты 
 
AT+CMGD=1CR  –  удаление сообщения в 1-м регистре SIM-карты 
 
 
В общем виде макрос обработки входящих SMS-сообщений должен состоять из следующих 
шагов: 
 
Шаг 1.  
Конвертация в коды ASCII команд чтения и удаления входящих сообщений. 
 
Шаг 2.  
Отправка в GSM-модем АТ-команды на чтение входящего сообщения в первом регистре SIM-
карты командой PUTCHARS. 
 
Шаг 3 
Чтение командой GETCHARS ответного сообщения от GSM-модема. 
 
Шаг 4.  
Вычленение из сообщения полезной информативной части для дальнейшей обработки. 
 
В ответ на АТ-команду чтения сообщения GSM-модем выдаст следующий ответ: 
 
CRLF+CMGR:_”REC_UNREAD”,”+79175138747”,,”09/02/24,15:45:00+12”CRLF!2CRLFCRL
F023!CRLF CRLFOKCRLF 

В данном примере было послано сообщение !2 (выделено) с телефона 89175138747.  
GSM-модем выдает ответ, содержащий ряд служебный полей: телефон отправителя, дату и 
время получения SMS, далее идет собственно сообщение.  
Особое внимание необходимо уделить полю между телефонным номером и датой, которое в 
данном примере пустое и ограниченно двумя запятыми, но может содержать некоторые 
символы, взятые из телефонной книги отправителя. В большинстве случаев оно остается 
пустым, но надо иметь его ввиду.  
 
Полезная часть в данном примере это «!2», а точнее цифра «2», которую необходимо 
переместить для дальнейшей обработки. Если начальным регистром под ответное сообщение 
в команде GETCHARS задан регистр $400, то «!2» окажется в регистре $431, причем в 
перевернутом виде. Т.е. в регистре $431 будет значение «3221H». 
 
Чтобы вычленить цифру «2» как полезную часть сообщения, необходимо сделать следующее: 
 
$610 = $431 
SWAP($611, $610, 1) 
$612 = W2B($611, 1) 
 
После исполнения вышеприведенного блока в регистре $612 будет значение 0032Н, что в 
дальнейшем можно использовать в командах сравнения, по результатам которых 
осуществить заданные действия.  
 
При повторном чтении слово UNREAD будет заменено на READ, следовательно «!2» 
сместиться в регистр $430. Данный факт необходимо учитывать при написании макроса (хотя 
при правильном программировании такой ситуации возникать не должно).  
 
Шаг 5 
Удаление SMS-сообщения в 1-м регистре SIM-карты. 
 
Шаг 6 
Очистка зоны приема SMS, а также регистра инструкции сравнения.  
 
Данную процедуру можно осуществить командой BMOV, скопировав нулевые регистры в 
зону приема SMS: 
 
BMOV($400, $800, 99) 
BMOV($610, $800, 3) 
 
(подразумевается, что в диапазоне регистров $800 - $899 находятся нули). 
 
Шаг 7 
Очистка буферов отправки и приема СОМ порта командой CLEARCOMBUFFER.  
 
 
Ниже приводятся примеры макросов, реализующих вышеприведенные действия. 

Чтение входящей SMS из 1-го регистра SIM-карты: 
 
CHR($500, "AT+CMGR=1 ") 
$505 = 000DH 
CLEARCOMBUFFER(0, 0) 
CLEARCOMBUFFER(0, 1) 
$702 = PUTCHARS($500, 12, 500) 
$703 = GETCHARS($400, 70, 500) 
 
 
Удаление SMS из 1-го регистра SIM-карты и очистка зоны приема и сравнения: 
 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($507, "AT+CMGD=1 ") 
$512 = 000DH 
$704 = PUTCHARS($507, 12, 500) 
BMOV($400, $800, 99) 
BMOV($610, $800, 3) 
CLEARCOMBUFFER(0, 0) 
CLEARCOMBUFFER(0, 1) 
 
 
 
Одновременные отправка и прием SMS-сообщений панелью оператора  
 
 
Макросы в панели обрабатываются построчно, т.е. сразу же осуществляется действие, 
записанное в данной строке макроса. Таким образом, конфликтов из-за одновременного 
использования порта не возникает. Все команды на отправку или прием данных из СОМ 
порта будут обрабатываться последовательно. Для этого необходимо лишь фиксировано 
включить условие данного события, а после его отработки обязательно сбросить. Тогда 
макрос будет работать циклически, построчно обрабатывая все команды, не сталкиваясь с 
конфликтами, и не пропуская ни одного запроса.  
 
 
Ниже в качестве примера приведен рабочий коммуникационный макрос, работающий по 
следующему алгоритму через СОМ1 (RS232): 
 
При включении бита $1000.3 отправляется аварийное сообщение «Протечка!» 

При включении бита $1000.4 отправляется аварийное сообщение «Отключили свет!» 

При получении СМС «!1» включается бит $1000.1, на панели загорается индикатор 
«Обогрев ВКЛ» и посылается СМС с сообщением «Обогрев ВКЛ» 
 

При получении СМС «!3» выключается бит $1000.1, на панели загорается индикатор 
«Обогрев ВЫКЛ» и посылается СМС с сообщением «Обогрев ВЫКЛ» 
 

При получении СМС «!2» включается бит $1000.2, на панели загорается индикатор «Насос 
ВКЛ» и посылается СМС с сообщением «Насос ВКЛ» 

При получении СМС «!4» выключается бит $1000.2, на панели загорается индикатор «Насос 
ВЫКЛ» и посылается СМС с сообщением «Насос ВЫКЛ» 
 
 

 
 
 
 
#Разовая установка параметров IRA 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($80, "AT+CSCS="IRA" ") 
$87 = 000DH 
CHR($60, "AT+CSMP=17,167,0,241") 
$70 = 000DH 
$111 = PUTCHARS($80, 16, 500) 
Delay(1000) 
$170 = PUTCHARS($60, 22, 500) 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
#Разовое удаление СМС из памяти SIM 
SELECTCOM(0) 
CHR($507, "AT+CMGD=1 ") 
$512 = 000DH 
$704 = PUTCHARS($507, 12, 500) 
BMOV($400, $800, 99) 
# Начало постоянно исполняемой части макроса 
LABEL 5 
#Проверка условия отправки СМС Протечка! 
IF $1000.3 == ON 
# Установка параметров UCS2 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($170, "AT+CSCS="UCS2"") 



стр. 14 из 20 

$177 = 000DH 
CHR($190, "AT+CSMP=17,167,0,25 ") 
$200 = 000DH 
$280 = PUTCHARS($170, 16, 500) 
Delay(1000) 
$300 = PUTCHARS($190, 22, 500) 
#Преобразование в UNICODE тел. номера и текста "Протечка!" 
CHR($210, "AT+CMGS="00380039003100360033003700360039003800330031"") 
$237 = 000DH 
CHR($250, "041F0440043E044204350447043A04300021") 
$268 = 001AH 
#Отправка АТ-команды и телефонного номера 
$320 = PUTCHARS($210, 56, 500) 
Delay(1000) 
#Отправка текста "Протечка!" 
$340 = PUTCHARS($250, 38, 500) 
CLRB $1000.3 
ENDIF 
#Проверка условия для отправки СМС "Отключили свет!" 
IF $1000.4 == ON 
# Установка параметров UCS2 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($170, "AT+CSCS="UCS2"") 
$177 = 000DH 
CHR($190, "AT+CSMP=17,167,0,25 ") 
$200 = 000DH 
$280 = PUTCHARS($170, 16, 500) 
Delay(1000) 
$300 = PUTCHARS($190, 22, 500) 
#Преобразование в UNICODE тел. номера и текста "Отключили свет!" 
CHR($210, "AT+CMGS="00380039003100360033003700360039003800330031"") 
$237 = 000DH 
CHR($250, "041E0442043A043B044E04470438043B0438002004410432043504420021") 
$280 = 001AH 
#Отправка АТ-команды и телефонного номера 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
$320 = PUTCHARS($210, 56, 500) 
Delay(1000) 
#Отправка текста "Отключили свет!" 
$340 = PUTCHARS($250, 62, 500) 
CLRB $1000.4 
ENDIF 
#Проверка условия включения индикатора и отправки СМС "Обогрев ВКЛ" 
IF $1000.5 == ON 
SETB $1000.1 
# Установка параметров UCS2 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($170, "AT+CSCS="UCS2"") 
$177 = 000DH 

CHR($190, "AT+CSMP=17,167,0,25 ") 
$200 = 000DH 
$280 = PUTCHARS($170, 16, 500) 
Delay(1000) 
$300 = PUTCHARS($190, 22, 500) 
#Преобразование в UNICODE тел. номера и текста "Обогрев ВКЛ" 
CHR($210, "AT+CMGS="00380039003100360033003700360039003800330031"") 
$237 = 000DH 
CHR($250, "041E0431043E043304400435043200200412041A041B") 
$272 = 001AH 
#Отправка АТ-команды и телефонного номера 
$320 = PUTCHARS($210, 56, 500) 
Delay(1000) 
#Отправка текста "Обогрев ВКЛ" 
$340 = PUTCHARS($250, 46, 500) 
CLRB $1000.5 
ENDIF 
#Проверка условия включения индикатора и отправки СМС "Обогрев ВЫКЛ" 
IF $1000.6 == ON 
CLRB $1000.1 
# Установка параметров UCS2 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($170, "AT+CSCS="UCS2"") 
$177 = 000DH 
CHR($190, "AT+CSMP=17,167,0,25 ") 
$200 = 000DH 
$280 = PUTCHARS($170, 16, 500) 
Delay(1000) 
$300 = PUTCHARS($190, 22, 500) 
#Преобразование в UNICODE тел. номера и текста "Обогрев ВЫКЛ" 
CHR($210, "AT+CMGS="00380039003100360033003700360039003800330031"") 
$237 = 000DH 
CHR($250, "041E0431043E043304400435043200200412042B041A041B") 
$274 = 001AH 
#Отправка АТ-команды и телефонного номера 
$320 = PUTCHARS($210, 56, 500) 
Delay(1000) 
#Отправка текста "Обогрев ВЫКЛ" 
$340 = PUTCHARS($250, 50, 500) 
CLRB $1000.6 
ENDIF 
#Проверка условия включения индикатора и отправки СМС "Насос ВКЛ" 
IF $1000.7 == ON 
SETB $1000.2 
# Установка параметров UCS2 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($170, "AT+CSCS="UCS2"") 
$177 = 000DH 
CHR($190, "AT+CSMP=17,167,0,25 ") 
$200 = 000DH 

$280 = PUTCHARS($170, 16, 500) 
Delay(1000) 
$300 = PUTCHARS($190, 22, 500) 
#Преобразование в UNICODE тел. номера и текста "Насос ВКЛ" 
CHR($210, "AT+CMGS="00380039003100360033003700360039003800330031"") 
$237 = 000DH 
CHR($250, "041D04300441043E044100200412041A041B") 
$268 = 001AH 
#Отправка АТ-команды и телефонного номера 
$320 = PUTCHARS($210, 56, 500) 
Delay(1000) 
#Отправка текста "Насос ВКЛ" 
$340 = PUTCHARS($250, 38, 500) 
CLRB $1000.7 
ENDIF 
#Проверка условия включения индикатора и отправки СМС "Насос ВЫКЛ" 
IF $1000.8 == ON 
CLRB $1000.2 
# Установка параметров UCS2 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($170, "AT+CSCS="UCS2"") 
$177 = 000DH 
CHR($190, "AT+CSMP=17,167,0,25 ") 
$200 = 000DH 
$280 = PUTCHARS($170, 16, 500) 
Delay(1000) 
$300 = PUTCHARS($190, 22, 500) 
#Преобразование в UNICODE тел. номера и текста "Насос ВЫКЛ" 
CHR($210, "AT+CMGS="00380039003100360033003700360039003800330031"") 
$237 = 000DH 
CHR($250, "041D04300441043E044100200412042B041A041B") 
$270 = 001AH 
#Отправка АТ-команды и телефонного номера 
$320 = PUTCHARS($210, 56, 500) 
Delay(1000) 
#Отправка текста "Насос ВЫКЛ" 
$340 = PUTCHARS($250, 42, 500) 
CLRB $1000.8 
ENDIF 
LABEL 4 
#Установка параметров IRA 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($80, "AT+CSCS="IRA" ") 
$87 = 000DH 
CHR($60, "AT+CSMP=17,167,0,241") 
$70 = 000DH 
$111 = PUTCHARS($80, 16, 500) 
Delay(500) 
$170 = PUTCHARS($60, 22, 500) 
Delay(500) 

# Отправка АТ-команды на чтение входящей СМС в 1-м регистре SIM-карты и получение 
ответа 
CHR($500, "AT+CMGR=1 ") 
$505 = 000DH 
CLEARCOMBUFFER(0, 0) 
CLEARCOMBUFFER(0, 1) 
$702 = PUTCHARS($500, 12, 500) 
$703 = GETCHARS($400, 70, 500) 
# Извлечение полезной составляющей СМС 
$610 = $431 
SWAP($611, $610, 1) 
$612 = W2B($611, 1) 
# Блок сравнения СМС с условиями включения заданных действий 
IF $612 == 0 
GOTO LABEL 1 
ENDIF 
# Проверка условия действия "Включить обогрев" и отправки СМС "Обогрев ВКЛ" 
IF $612 == 31H 
SETB $1000.5 
GOTO LABEL 2 
ENDIF 
# Проверка условия действия "Включить насос" и отправки СМС "Насос ВКЛ" 
IF $612 == 32H 
SETB $1000.7 
GOTO LABEL 2 
ENDIF 
# Проверка условия действия "Выключит обогрев" и отправки СМС "Обогрев ВЫКЛ" 
IF $612 == 33H 
SETB $1000.6 
GOTO LABEL 2 
ENDIF 
# Проверка условия действия "Выключит насос" и отправки СМС "Насос ВЫКЛ" 
IF $612 == 34H 
SETB $1000.8 
GOTO LABEL 2 
ENDIF 
# Блок оценки некорретных СМС и переход к их удалению 
IF $612 > 34H 
GOTO LABEL 2 
ENDIF 
IF $612 < 31H 
GOTO LABEL 2 
ENDIF 
# АТ-команда на удаление СМС в 1-м регистре SIM-карты 
LABEL 2 
$130 = INITCOM(0, 0, 1, 0, 0, 6, 0) 
SELECTCOM(0) 
CHR($507, "AT+CMGD=1 ") 
$512 = 000DH 
$704 = PUTCHARS($507, 12, 500) 
BMOV($400, $800, 99) 
BMOV($610, $800, 3) 

CLEARCOMBUFFER(0, 0) 
CLEARCOMBUFFER(0, 1) 
LABEL 1 
# Переход в начало постоянно исполняемой части макроса 
GOTO LABEL 5 
END 

Приложение 1 

Приложение 2 
 
 
 

Кириллические символы в UNICODE 
 
 

 
 
 
Например: 
 
А – Н0410 
а –  Н0430 
Т – Н0422 
п –  Н043F 
 
 
 
 
 
 
 
 
 


